Appendix A3. Partial Differential Equations

The wave equation

The following entries show how *Mathematica* can be used to plot approximations to solutions of the wave equation on a finite domain. See Simmons/Krantz, Chapter 6, Section 2.

Consider a string of length L stretched along the x-axis with ends clamped at $x = 0$ and $x = L$. Let $y(x, t)$ denote the vertical position at time t of the point on the string that is directly above (or below) the point $(x, 0)$. For small vibrations, y satisfies the wave equation

$$y_{tt} = a^2 y_{xx}$$

The letter a is a positive constant determined by the characteristics of the string. Separation of variables leads to solutions of the following form

$$Y_N(x, t) = \sum_{j=1}^{N} \left(a_j \cos\left(\frac{a j \pi t}{L}\right) + b_j \sin\left(\frac{a j \pi t}{L}\right)\right) \sin\left(\frac{j \pi x}{L}\right), \; N \text{ a positive integer.}$$

Set the string into motion

The string is set into motion at $t = 0$ by giving it an initial shape $f(x)$ and an initial velocity distribution, $g(x)$. Thus the coefficients a_j and b_j should be chosen so that the function

$$Y_N(x, 0) = \sum_{j=1}^{N} a_j \sin\left(\frac{j \pi x}{L}\right)$$

approximates $f(x)$ on $[0, L]$ and the function

$$\partial_t Y_N(x, 0) = \sum_{j=1}^{N} \frac{a j \pi b_j}{L} \sin\left(\frac{j \pi x}{L}\right)$$

approximates $g(x)$. Consequently, a_j is the Fourier sine series coefficient for $f(x)$ and $\frac{a j \pi b_j}{L}$ is the Fourier sine series coefficient for $g(x)$.

The following entries define the functions f and g, calculate a_j and b_j, then create various solution curves. We assume that $L = 1$, $a = 1$ and the string is initially stretched "tent like" over the x axis with the shape

$$f(x) = 0.2\,x + (0.2\,(1 - x) - 0.2\,x)\,\text{UnitStep}(x - 0.5)$$

See the following definitions and plot.

In[1]:=
```
L = 1; a = 1;
f[x_] := 0.2*x + (0.2*(L-x)-0.2*x)*UnitStep[x-0.5];
Plot[ f[x], {x,0,1}, PlotRange->{0,0.125}, AspectRatio->1/3 ]
```



Set the string into motion with a finger flick at a point one quarter of the way from the left endpoint

$$g(x) = 0.1 \text{ DirecDelta}(x - 0.25)$$

In[4]:=
```
g[x_] := 0.1*DiracDelta[x - 0.25]
```

You may, of course, change these to fit any situation that you would like to explore.

The next entries calculate the formulas for the coefficients aj and bj.

In[15]:=
```
aj = 2/L*Integrate[f[x]*Sin[j*Pi*x/L], {x,0,L} ]
bj = L/(a*j*Pi)*2/L*Integrate[g[x]*Sin[j*Pi*x/L], {x,0,L} ]
```

Out[15]= $\frac{2\,(0.0405285\,\mathrm{Sin}[1.5708\,j] - 0.0202642\,\mathrm{Sin}[j\,\pi])}{j^2}$

Out[16]= $\frac{0.063662\,\mathrm{Sin}[0.785398\,j]}{j}$

This is the definition of the function *Y* as a function on *N, x, t* :

In[17]:=
```
Y[N_,x_,t_] := Sum[ (aj*Cos[a*j*Pi*t/L] + bj*Sin[a*j*Pi*t/L])*Sin[j*Pi*x/L],
{j,1,N}]
```

The first plot checks that the coefficients are correct for the velocity function g. (A check for the shape function *f* is made when we plot *Y* at *t* =0 below).

In[24]:=
```
Plot[ {g[x], Sum[ a*j*Pi*bj/L*Sin[j*Pi*x/L], {j,1,30}]}, {x,0,L},
PlotRange->{-0.8,3.5}, PlotLabel->"Initial Velocity"]
```



This curve is a typical approximation to a Dirac delta. The area under the curve is approximately 1/10.

In[23]:= **Integrate[Sum[a * j * Pi * bj / L * Sin[j * Pi * x / L], {j, 1, 30}], {x, 0, 1}]**

Out[23]= 0.100099

The following plot of $Y(20,x,0)$ shows that the aj coefficients are also correct.

```
In[26]:= Plot[ Y[20,x,0], {x,0,L}, PlotRange->{0,0.12}, PlotLabel->"Initial
Waveform", AspectRatio->1/3]
```



A snapshot of the waveform at $t = 0.1$.

```
In[28]:= Plot[ Y[50,x,0.1], {x,0,L}, AspectRatio->1/3]
```



Five snapshots, one every 0.2 seconds:

```
In[29]:= Plot[ Evaluate[Table[Y[50,x,t], {t,0,1,0.2}]], {x,0,L}]
```



A movie (see the Help Browser: A Practical Introduction to *Mathematica*/Graphics and Sound/Special Topic: Animated Graphics).

```
In[32]:= <<Graphics`Animation`
Animate[Plot[Y[50,x,t], {x,0,L}, PlotRange->{-0.2,0.2}], {t,0,2,0.05}]
```



The waveform surface

In[34]:= `Plot3D[ Y[50,x,t], {x,0,L}, {t,0,2}, ViewPoint->{1.542, -2.913, 0.764} ]`